
要旨
ベイズモデルによる統計解析を行う際、SASプログラムにWinBUGSの制御を組み込むことでMCMC法による事後分布の導出部分のみをWinBUGSに行わせ、全体の処理フローはSASで行う方法を紹介する。

1. はじめに
1.1 ベイズ統計学とMCMC法
　近年、医薬統計分野におけるベイズ分析の実例が増えつつある。ベイズ分析がより身近なものとして受け入れられつつある理由として、ベイズ分析のための数値計算手法が確立され、実際のデータへの適用が容易となったことが挙げられる。
ベイズモデルによるデータ解析を実際に行う場合、その事後分布に興味のあるところだが、事後分布導出のための計算には、多次元の確率密度関数の積分が頻繁に登場し、数値積分が通常の数値解析手法では得られにくい場合がある。現在では、解析的には困難な事後分布の導出にはMCMC(Markov chain Monte Carlo)法が標準的に用いられている[1]。
MCMC法はマルコフ連鎖の性質を利用して、任意の確率分布から乱数を生成するマルコフ連鎖サンプリング法を応用したモンテカルロ法である。MCMC法は1950年代に核物理学の分野で開発されたモンテカルロ法で、1980年代にベイズ分析に応用可能であることが知られることとなり、以後、ベイズ統計学においてその適用方法が発展し、ベイズ分析の実問題への適用に大きく貢献している[1]。
1.2 MCMC法とWinBUGS
SASでは、 V9.2からMCMCプロシジャが評価版として実装され、MCMC法を用いたベイズ分析が可能となったが、これまで、MCMC法を用いてベイズ分析を行う場合に一般に広く用いられてきたものはWinBUGSである[2]。SASに並び広く用いられている解析ソフトRでは、RのMCMCパッケージを用いてMCMC法による計算を行うことが可能である。しかし、RからWinBUGSを制御し、MCMC法を実装する方法(R2WinBUGS package)が採用されていることも多い [2]。また、ベイズ分析を用いた臨床試験においても、MCMC法による計算にWinBUGSを用いた例[3]もあるなど、WinBUGSの使用例は多い。
WinBUGSを用いると、MCMC法の実装のために、特に専門的な知識や複雑なプログラムを書く必要もなく、ギブスサンプラーによるMCMC法を容易に実装する事ができる。しかし、WinBUGSには、SASなどの様にデータハンドリングを行うための機能が用意されていない。また、Loopや条件分岐などの仕組み自体も用意されていない。よって、複数のデータを用いた解析やシミュレーションを行う場合、WinBUGSのみでは扱いづらい。また、WinBUGSのインターフェイスに用意されたタブやボタンを手動で操作して得られた解析結果には、SASプログラムの様にパッケージとしての再現性を求めることが難しい。
　本稿では、Zhang(2008)[4]で紹介された方法を参考に、SASによってデータのハンドリングやデータ解析全般の処理の制御を行い、SASからWinBUGSを制御することで、MCMC法による事後分布の導出の計算部分のみをWinBUGSに行わせ、その後の計算結果のハンドリングについてはSASを用いて行う方法を紹介する。これにより複数の解析対象データやデータハンドリングの必要なデータに対しても、より容易にベイズ分析を行うことが可能となる。
　以後、2章においてMCMC法とそのアルゴリズムであるギブスサンプラーについて簡単に紹介し、3章において、SASを用いてWinBUGSを操作する方法を紹介する。また、4章においてSASを用いてWinBUGSを繰り返し実行することでベイズモデルのモンテカルロシミュレーションを実行する方法を紹介する。
2. MCMC法とギブスサンプラー
　本章では、ベイズ分析とMCMC法の関わりと、MCMCサンプリングの良く知られたアルゴリズムであり、WinBUGSに実装されているギブスサンプラーについて紹介する。
2.1 ベイズ分析とMCMC法
　ベイズ分析では、データが得られる前の事前情報から構成したパラメータの事前分布を、得られたデータによって更新し、事後分布を得ることで統計的推測を行う。
パラメータ

　ベイズ統計学においては事後分布

が多次元であることも多く、その場合、事後分布に基づく推測のために数値積分を求めることは難しい。そのため、事後分布からのモンテカルロ標本

をMCMC法によってサンプリングし、得られたモンテカルロ標本から事後分布に関する要約や推測を行う。マルコフ連鎖には適当な初期値からはじめて、十分な推移の回数を重ねていくと、マルコフ連鎖からの標本が不変分布に収束していくという性質がある。この不変分布が事後分布

となるようにマルコフ連鎖を構成することで、マルコフ連鎖からの標本

からのモンテカルロ標本として扱うことが出来る[5]。
2.2 ギブスサンプラー
　ギブスサンプラーはマルコフ連鎖の推移核を構築する一般的な方法としてよく知られており、MCMC法によるサンプリングを行うためにWinBUGSに実装されているアルゴリズムである。
　目標とする事後分布

のように、いくつかのブロックに分割可能であるとする。また、

が与えられたときの条件付き事後分布

とし、この条件付き分布からのサンプリングが容易であると仮定する。このとき、ギブスサンプラーは以下の様なアルゴリズムとなる[5]。
(1) 初期値

(3) 

(2)、(3)を繰り返し、十分大きな数

のモンテカルロ標本とする。
3. SAS-WinBUGSを用いたMCMC法の実装
3.1 処理の流れ
　ここでは、SASプログラムを用いてWinBUGSを制御することにより、ベイズモデルのMCMC法による推定を行う方法を紹介する。
以下の図3.1の様な流れで、SASによってデータのハンドリング及びLoopや条件分岐などのデータ解析全体の処理の制御を行い、MCMC法による事後分布の計算部分だけをWinBUGSに行わせ、事後分布の計算結果をSASに返し、SASを用いて計算結果をハンドリングするSASプログラムを用意する。本稿ではWinBUGSをSASプログラムの一部の様に扱うことでSASを用いてベイズモデルの推定を行う。

図3.1 SASによるWinBUGSの制御フロー
　本稿で紹介するSASプログラムを用いて解析を行うためのフローを順に詳しく説明していく。例として、Smith(2006)[3]で用いられたNDLM(Normal Dynamic Linear Model)による用量反応曲線の推定を考える。NDLMは状態空間モデルの特殊な場合で、単調性の仮定が必要なく、Smith(2006)[3]の他にも用量反応曲線の推定に用いられた実例がある[6][7]。
	例3.1   単調性の仮定出来ない用量反応曲線のNDLMによる推定
  
[image: image32.wmf]j

Y

を投与群jにおけるDose response(平均値)とする。
また、投与群j=1はPlacebo群(Dose responseなし)を表すこととする。Dose response curveには
以下の様な状態空間表現を考える。
観測方程式：    

システム方程式：

また、ここでは、パラメータには事前情報を仮定しないこととし、事前分布がそれぞれ自然共役事前分布となる様に、

には逆ガンマ分布をそれぞれ用いることとした。また、Smith(2006)[3]より、

には一様分布を仮定する。各パラメータに仮定する事前分布については(3.3)に示す。

興味があるのは 、投与群jのDose response 

の事後分布である。また、本例では

について
以下の様なデータが観測されているとし、データに基づいてNDLMのパラメータを推定する。
観測されたDose responseの値

手順① データハンドリング
SASプログラムを用いて解析対象データのハンドリングを行う。例えば、臨床試験のデータの場合であれば、特定の解析対象集団に属する被験者のデータに絞り、臨床検査値の前値と後値の差を求めるなどのハンドリングを行うことが考えられる。
手順② モデリング
統計モデルを記述するBUGSコードは非常に簡潔であり、尤度と事前分布を記述するだけでMCMC法による事後分布の導出を行える。WinBUGSでは、BUGSコードをtextファイルとして読み込むことでモデルに沿った解析を行う。ここでは例3.1のNDLMをBUGSコードでプログラミングし、textファイルとしてSASプログラムから読み取り可能な任意の場所へ保存する処理を行う。NDLMのBUGSコードはSmith(2006)[3]を参考にする。
SASプログラム3.1　　　BUGSコードで記載した統計モデルをtextファイルとして保存
	*モデルの格納をしておく;

data model;

  input model $80.;

  cards;

###　　　以下のBUGSコードで記載したモデルをtextファイルとして保存　　　　　　###

model{

#尤度を記述する
#dnorm()には平均と分散の逆数を指定する
for(j in 1:N){

y[j] ~ dnorm(mu[j], sigma2inv)

mu[j] <- theta[j]

}

for(k in 2:N){

theta[k]~dnorm(mu.theta[k], prec.theta[k])

mu.theta[k]<- theta[k-1] + delta[k-1]

delta[k]~dnorm(delta[k-1], prec.delta[k])

prec.theta[k]<- 1/(W * sigma * sigma)

prec.delta[k]<- 1/(W * sigma * sigma)

}

#事前分布を設定する
theta[1]~dnorm(0,1.0E-4)

delta[1]~dnorm(0, 1.0E-4)

beta ~ dnorm(0,1.0E-4)

sigma2inv ~ dgamma(0.1, 0.1)

sigma <- 1/sqrt(sigma2inv)

W ~ dunif(0.01,100)

};

##############　　　　BUGSコードによる統計モデル　　　　　　　　################

run

*モデルをtextファイルとして保存;

data _null_;

  set model;　　　　　　　　　　　*フォルダ「WinBUGS」は予め作成しておく;

    file "C:\WinBUGS\NDLMmodel.txt";    *左のパスにBUGSコードによる統計モデルを保存する;

  put model;

run;


※WinBUGSの出力結果や統計モデル、解析データを保存するフォルダ「C:WinBUGS」を予め作成しておく。
手順③ WinBUGS形式のデータセットを作成
　WinBUGSに解析対象となるデータを読み込むためには、SASプログラムでハンドリングしたSASデータセットをWinBUGSで読み込むことが可能なデータ形式でtextファイルとして任意の場所に保存する必要がある。ここでは、本稿で作成したSASマクロ%DTRANS（Appendix A参照）を用いて、図3.2の様にSASデータセットをWinBUGSデータセットに変換する処理を行う。SASプログラム3.2の様に%DTRANSのマクロ引数にWinBUGSデータセットに変換するSASデータセット名、変換後のファイルパス、変換する変数名を記載すると、図3.2の様にデータの変換が実行される。
SASプログラム3.2   %DTRANSによるWinBUGSデータへの変換
	%dtrans(data=data,  outfile=C:\WinBUGS\Data.txt,  var=y);

図3.2 SASデータセットのWinBUGSデータセットへの変換
手順④ MCMCによるパラメータ推定
　MCMC法によるパラメータ推定の処理をSASプログラムからWinBUGSに命令を与えることで、自動で行う。WinBUGSは通常、インターフェイスに用意されたタブやボタンを操作することで処理を行うが、バッチ処理で命令を与えれば、SASプログラムからも処理手順の命令を与えることが出来る。ここでは、SASプログラム3.3を用いてSASプログラムからバッチ処理でWinBUGSに命令を与える。
　以下では、WinBUGSに統計モデルの推定を行わせるバッチファイルを作成する。WinBUGSのコマンドはWinBUGSマニュアル[8]に整理されており、ここではマニュアル[8]を参考にバッチファイルの作成を行う。
SASプログラム3.3  WinBUGSを制御するバッチファイルの作成
	data _null_;

  file "C:\program files\WinBUGS14\NDLMBatch.txt";

  put // @@
#1  "display('log')"                           実行logを保存する
#2  "check('C:/WinBUGS\NDLMmodel.txt')"　　　コード3.1のモデルの読み込み&構文チェック
#3  "data('C:/WinBUGS\Data.txt')"　　　WinBUGSデータ3.1のデータを読み込む
#4  "compile(3)"                                  統計モデルをコンパイルする
#5  "gen.inits()"　　　　　　　　　　　　　　　MCMCの初期値を乱数によって生成
#6  "update(1000)"          Burn inの数を設定  -MCMCの系列の最初の1000個のデータは捨てる-

#7  "thin.updater(20)"　　　MCMC stepのサンプリング間隔を指定  -サンプリング間隔を20に設定-
#8  "set(mu)"       Setで事後分布を計算するパラメータを指定する  -
[image: image48.wmf]q

の事後分布を推定-

#9  "set(W)"　　　　 - 
[image: image49.wmf]W

の事後分布を計算-

#10 "set(delta)"       - 
[image: image50.wmf]d

の事後分布を計算-

#11 "set(sigma2inv)"   - 
[image: image51.wmf]2

s

の事後分布を計算-

#12 "dic.set()"                    DIC fit statisticsを出力
#13 "update(1000)"                MCMC法のサンプリング数を設定  -1000に指定-

#14 "dic.stats()"
#15 "coda(*,'C:/WinBUGS\MCMCout')"     計算結果をMCMCoutというファイル名で出力
#16 "save('C:/WinBUGS\log.odc')"　　　　  WinBUGSの出力outputを保存
#17 "quit()"                             WinBUGSを終了
run;
*上記のスクリプトをバッチコード化して保存;

data _null_;

  file "C:\WinBUGS\NDLM.bat";

  put  "CD C:\program files\WinBUGS14";

  put "WinBUGS14.exe /PAR NDLMBatch.txt";

  put "exit";

run;


上記コードのコマンドをいくつか補足で説明すると、#4ではSASプログラム3.1にて記述した統計モデルをコンパイルし、MCMCのチェーンの数を指定する。チェーンの数を増やせば処理が遅くなるが、チェーンの数を2以上に設定すると、マルコフ連鎖が不変分布に収束しているかをGelman-Rubin統計量を用いてチェックすることが出来る。しかし、マシンスペックに不安がある場合や、同一の設定でマルコフ連鎖が不変分布に収束していることを確かめているなどの理由がある場合はチェーンの数を1にすることも検討して良い[2]。
　WinBUGSにより初期条件が与えられ、マルコフ連鎖の生成が開始される。初期条件に依存していると考えられるモンテカルロ標本は推定に用いず、廃棄する。このことをBurn inというが、#6では、Burn inで廃棄するモンテカルロ標本の個数を決定する。一般にBurn inで廃棄されるサンプル数は大きいほど安全であるが、具体的にサンプル数をいくつに設定するのが良いという基準はない。本例では1000と設定している。
　#7では、マルコフ連鎖からモンテカルロ標本をサンプリングするサンプリング間隔を決定している。マルコフ性による条件付き独立は仮定できるが、定常確率過程の自己相関があるため、モンテカルロ標本は自己相関が弱まり独立なサンプルと考えられる程度のサンプリング間隔を設ける必要がある。
#13では、マルコフ連鎖からモンテカルロ標本をサンプリングするサンプル数を決定する。ここでのサンプル数は#7で決定したサンプリング間隔を空けてサンプリングされるモンテカルロ標本のサンプル数である。
#15ではマルコフ連鎖から採集したモンテカルロサンプルをtextファイルとして保存する。本例では、モンテカルロ標本のファイル名を「MCMCout」としたため、#4で指定したチェーンの数が3の場合、3つのtextファイル(MCMCout1.txt、MCMCout2.txt、MCMCout3.txt)それぞれに、#13で指定したサンプル数だけモンテカルロ標本が保存される。
また#16では、WinBUGSで出力したGelman Rubin統計量のグラフなどのoutputを保存する。
手順⑤ MCMCによる計算結果を出力
　SASプログラム3.3で作成されたバッチファイルを実行することによって、WinBUGSに一連の解析処理を行わせることが出来る。バッチファイルを実行して得られたモンテカルロ標本は図3.3右の様な形式で、SASプログラム3.3の#15で指定したフォルダパスに出力される。ここでは、WinBUGSから出力されたモンテカルロ標本を、SASでハンドリングが行い易い様にパラメータごとに変数を作成し、SASデータセット化する。
WinBUGSのモンテカルロ標本出力は、値が一列に出力されているのみで、この状態では、どのレコードがどのパラメータのモンテカルロ標本であるかを特定することが出来ない。WinBUGSは#15で指定したフォルダパスに図3.3左の様なモンテカルロ標本出力のレコードとパラメータとを紐づけるINDEXファイルを出力するので、INDEXファイルを用いてパラメータとモンテカルロ標本の関係を紐づける。また、INDEXファイルもモンテカルロ標本のファイルと同様に#15で指定したフォルダパスに出力される。本例では、INDEXファイルのファイル名は「MCMCoutindex.txt」となる。

図3.3 パラメータINDEXとモンテカルロ標本
　上記のINDEXファイルの1行目には「W    1    1000」と記載されている。これはモンテカルロ標本ファイルの中で、1行目から1000行目までがパラメータ
[image: image53.wmf]W

のモンテカルロ標本であることを表している。同様にパラメータdelta[1]のモンテカルロ標本は1001行目から2000行目までであることを表す。本稿ではこのパラメータINDEXファイルをもとにモンテカルロ標本ファイルをパラメータごとに変数を作成し、SASデータセットに変換するマクロ%CODATRANS（Appendix B参照）を用いる。
%CODATRANSはSASプログラム3.4の様に、マクロ引数codapathにWinBUGSによるモンテカルロ標本ファイルおよびINDEXファイルが保存されているフォルダパス、prefixにはSASプログラム3.3の#15で指定したファイル名、outdataにはSASデータ変換後のモンテカルロ標本のSASデータセット名を記載する。また、outnumsにはSASプログラム3.3の#4で指定したマルコフ連鎖のチェーンの本数を記載する。本例の様にチェーンの数が3の場合、図3.3右のモンテカルロ標本ファイルが3つ作成されるため、%CODATRANSによってSASデータセット化されるモンテカルロ標本のサンプル数は3つのファイルの合計となる。また、事後分布のグラフを出力する場合はoutgraphにYを指定する。
SASプログラム3.4   %CODATRANSによるWinBUGSデータへの変換
	%codatrans(codapath=C:\WinBUGS, prefix=mcmcout, outcoda=coda, outstats=stat, outnums=3, outgraph=N);


%CODATRANSを用いてSASデータ変換された結果は図3.4の様になる。モンテカルロ標本ファイルをSASデータセット化し、パラメータごとに変数を作成した。このデータセットを用いることでSASによるデータハンドリングや解析結果の加工を行うことが出来る。

図3.4 %CODATRANSによるモンテカルロ標本のSASデータセット化
手順⑥ 推測、グラフ化
%CODATRANS によって作成されたSASデータセットCODAの各変数に格納された値はパラメータの事後分布からのモンテカルロ標本である。すなわち、SASデータセットCODAの各変数の度数グラフを作成すれば事後分布のグラフが得られる。また、%CODATRANSではマクロ引数outstatsにデータセット名を指定すると、指定したSASデータセットにCODAの各変数の要約統計量を算出することが出来る（図3.5参照）。
本例の目的はNDLMによるDose response curveの推定であるから、
[image: image54.wmf]j

の信用区間のグラフを作成すること
を考える。グラフは%CODATRANSによって算出されたSASデータセットの平均およびパーセント点を用いて行える。ここでは、GPLOTプロシジャを用いてDose response curveの事後分布からの推定値と80%信用区間を図3.6に図示する。
[image: image60.emf]以上の様な流れでMCMC法によるモンテカルロ標本のサンプリング部分のみをWinBUGSに行わせ、データ解析全体の処理はSASプログラムで行う。本例では、MCMC法によって計算された事後分布を図3.6の様に図示したが、行う解析の内容次第で手順⑥の処理内容は変化する。手順⑥の処理については本稿で紹介するSASプログラムを利用する方々がそれぞれの目的にあった処理を行われたい。
図3.5 %CODATRANSによる事後分布の要約統計量の算出

図3.6 NDLMを用いたDose response curveの推定
4. SAS-WinBUGSを用いたモンテカルロシミュレーションの実行
3章のSASプログラムを反復して実行することでモンテカルロシミュレーションが実行可能となる。ここでは、SASプログラムをLoopによって反復実行し、モンテカルロシミュレーションを行うプログラムの例を紹介する。
4.1 SASマクロによるモンテカルロシミュレーションの例(Appendix D)
　モンテカルロシミュレーションを行う場合には、図3.1の手順①③④⑤をLoopによって繰り返し、最後にシミュレーション結果をまとめる処理を行えばよい。
例えば3章の例と同様のモデルで、Dose response curveの推定シミュレーションを行いたいとする。簡略化のために、シミュレーションで発生させるデータは図4.1のDose response curveの周りに分布する分散12の正規分布とする。このデータを用いて100回推定を繰り返すシミュレーションを行う場合、SASプログラムはAppendix Dの様になる。Appendix Dでは、Loopを行う度にシードを変更し、乱数による解析データの作成を行っている。また、%CODATRANSによってモンテカルロ標本を要約統計量の形でSASデータセット化し、100回計算された要約統計量100オブザベーションをシミュレーションによる標本として保存しておく。最後に要約統計量をシミュレーション実行回数で平均し、シミュレーションによる推定結果を得る。得られた結果をGPLOTプロシジャで図示したものが図4.2である。この様に、SASプログラムを用いてWinBUGSを制御することで、MCMC法による計算が必要なモデルについてもモンテカルロシミュレーションを行うことが可能になる。

図4.1 モンテカルロシミュレーション用生成データ（実線の周りに分散12の正規分布を発生）

図4.2 モンテカルロシミュレーションによるDose response curveの推定結果
5. おわりに
本稿では、WinBUGSをSASプログラムの一部の様に扱い、MCMC法を用いたベイズ分析を行うSASプログラムの紹介を行った。本稿で紹介したSASプログラムを用いることで、MCMC法によるベイズ分析においても、通常のSASでの解析と同様に、解析データのハンドリングから解析結果の加工までをSASを用いて行うことが可能となる。また、本稿で紹介したSASプログラムを反復実行することでベイズモデルに対するモンテカルロシミュレーションも実行可能となることを示した。さらに、本稿で紹介したプログラムをマクロ化して用いることで、層別解析の様に同一のモデルを複数の解析データに対して適用する必要がある場合にも対応が可能である。 

紹介したプログラムはモンテカルロ標本をSASデータセット化し、要約統計量を算出するところまで対応している。その後の解析結果の加工については、統計解析の目的に合わせて自由に変更が可能である。

Appendix A  %DTRANS  SASデータセットをWinBUGSデータへ変換するマクロ
	/*-----------------------------------------------------------------------                                                                                                                               

SAS Version : Ver.9.1.3  

data        : 解析データ
var         : 分析変数（複数の変数を指定可能）例）var=x y z;

outfile     : WinBUGS取り込み用txtファイル名（パス含む）
format      : 解析データの出力用フォーマット
rounddigit  : 四捨五入の桁
ls          : 指定したラインサイズ以内でカンマごとに値を区切る（50以上を指定する）
-----------------------------------------------------------------------*/

%macro dtrans(data=, var=, outfile=, format=best32, rounddigit=15, ls=50);

%let i=0; *初期化;

%do %until(&&var&i=);

%let i=%eval(&i+1);

%let var&i=%scan(&var, &i, %str( )); *var&i:varの各変数名;

%end;

%let varn=%eval(&i-1); *var:変数の個数;

  *解析データのオブザベーション数を取得;

  %let dsid = %sysfunc(open(&data));

  %let nobs = %sysfunc(attrn(&dsid, nobs)); 

  %let rc = %sysfunc(close(&dsid));

  *解析データの出力用フォーマット;

  proc format;

    value varf

    ., .a - .z, ._ = "NA"

    other           = [ &format.. ]

    ;

  run;

  *解析データをマクロ変数に格納;

  data _null_;

    set &data;

    array _var(*) &var;

    do i=1 to dim(_var);

      if .z < _var(i) then _var(i)=round(_var(i), 10**(-&rounddigit));

      call symputx("val" || trim(put(i, best.-l)) || "_" || put(_n_, best.-l), put(_var(i), varf.));

    end;

  run;

  *WinBUGS取り込み用txtファイル保存;

  data _null_;

    file "&outfile" ls=&ls;

    put "list(N=&nobs,";

    %do i=1 %to &varn;

      put "&&var&i = c(";

      %if 1 < &nobs %then %do;

        %do j=1 %to &nobs-1;

          put "&&val&i._&j," @;

        %end;

      %end;

      put "&&val&i._&nobs" ")" 

      %if &i = &varn %then %do; ")" %end; 

      %else %do; "," %end;

      ;

    %end;

  run;

%mend dtrans;


Appendix B  %CODATRANS  WinBUGS出力のモンテカルロ標本をSASデータセット化するマクロ
	/*-----------------------------------------------------------------------                                                                                                                               

SAS Version : Ver.9.1.3  

codapath    : indexファイル、outputファイル読み込みパス
prefix      : BUGSコードで設定したファイル名
             （論文 SASプログラム3.3 #15で指定したファイル名）
outcoda     : WinBUGSで生成されたモンテカルロ標本を格納するSASデータセット名
outstats    : パラメータの要約統計量を算出するSASデータセット名（指定した場合に算出）
outnums     : outputファイルの数を指定
outgraph    : ヒストグラムを生成（outgraph=Yの場合に出力） 

-----------------------------------------------------------------------*/

%macro codatrans(codapath=, prefix=, outcoda=, outstats=, outnums=1, outgraph=N);

  *outputファイルのファイル参照名 設定;

  filename in ( %do i=1 %to &outnums ;

                  "&codapath\&prefix.&i..txt" 

                %end;

              );

  *outputファイルのSASデータセット化; 

  data sasout;

    infile in dlm='09'x end=eof;

    do until(eof);

      input number value;

      output;

    end;

  run;  

  *indexファイルのSASデータセット化;

  data parameter;

    length parameter $100;

    infile "&codapath\&prefix.index.txt" dlm='09'x end=eof;

    do until(eof);

      input parameter $ firstnum lastnum;

      prm=lowcase(translate(parameter, "_ ", "[]"));

      prm_grp=lowcase(scan(parameter, 1, "["));

      if index(parameter, "[") > 0 then prm_seq=input(scan(parameter, 2, "[]"), best.);

      else prm_seq=1;

      output;

    end;

  run;

  *indexファイルの値をマクロ変数に格納;

  proc sql noprint;

    select count(prm) into :vnmcnt from parameter;

    select parameter into :lbl_1-:lbl_%trim(&vnmcnt) from parameter; *グラフラベル用;

    select prm into :vnm_1-:vnm_%trim(&vnmcnt) from parameter;       *変数名をマクロ変数に格納;

    select count(firstnum) into :fstcnt from parameter;

    select firstnum into :fst_1-:fst_%trim(&fstcnt) from parameter;  *開始列番号をマクロ変数に格納;

    select count(lastnum) into :lstcnt from parameter;

    select lastnum into :lst_1-:lst_%trim(&lstcnt) from parameter;   *終了列番号をマクロ変数に格納;

  quit;

  *outputファイルの値を加工（outputの数をLoop）;

  %let lstnm= %left(&vnmcnt);

  %do j=1 %to &outnums;

    data _&outcoda._&j;

      merge %do i=1 %to &vnmcnt;

              sasout(firstobs=%eval(&&lst_&lstnm * (&j-1) + &&fst_&i) 

                          obs=%eval(&&lst_&lstnm * (&j-1) + &&lst_&i) 

                       rename=(value=&&vnm_&i))

            %end;

            ;

    run;

  %end;

  *加工したデータを1データセットに結合;

  data &outcoda;

    set %do j=1 %to &outnums;

          _&outcoda._&j

        %end;

        ;

  run;

  *要約統計量の算出;

  %if &outstats ^= %then %do;

    %do i=1 %to &vnmcnt;

      proc univariate noprint data=&outcoda;

        var &&vnm_&i;

        %if %upcase(&outgraph) = Y %then %do; 

          histogram &&vnm_&i / kernel; 

        %end;

        output out=_stats_&i n=n mean=mean var=var std=std min=min max=max

                             range=range pctlpts=10 median=median pctlpts=90

                             kurtosis=kurtosis skewness=skewness pctlpre=p;

        label &&vnm_&i="&&lbl_&i";

      run;

      *要約統計量におけるラベル名をマクロ変数に格納;

      %if &i=1 %then %do;

        proc contents data=_stats_&i out=_con noprint;

        run;

        data _con;

          set _con;

          _label=left(kscan(label, 2, "の"));

        run;

        proc sql noprint;

          select count(name) into :statsvnmcnt from _con;

          select name into :statsvnm_1-:statsvnm_%trim(&statsvnmcnt) from _con;

          select _label into :statslbl_1-:statslbl_%trim(&statsvnmcnt) from _con;

        quit;

      %end;

    %end;  

    *要約統計量を1データセットに結合;

    data &outstats;

      set %do i=1 %to &vnmcnt;

            _stats_&i

          %end;

          ;

      *要約統計量ラベルの付与;

      %do i=1 %to &statsvnmcnt;

        label &&statsvnm_&i="&&statslbl_&i";

      %end;

    run;

    *要約統計量にパラメータ情報を結合;

    data &outstats;

      set parameter(keep=parameter prm_grp prm_seq);

      set &outstats;

    run;

  %end;  

  *workデータの削除;

  proc datasets nolist;

    delete _&outcoda: _stats: sasout _con parameter;

  run ;

  quit;  

%mend codatrans;


Appendix C  NDLMの推定
	*解析データ;

data data;

  do dose=1 to 8;

    select(dose);

      when(1) y=0.8;

      when(2) y=0.2;

      when(3) y=0.5;

      when(4) y=1;

      when(5) y=5;

      when(6) y=4;

      when(7) y=5.5;

      when(8) y=4.5;

      otherwise;

    end;

    output;

  end;

run;

*モデルの格納をしておく;

data model;

  input model $80.;

  cards;

model{

for(j in 1:N){

y[j] ~ dnorm(mu[j], sigma2inv)

mu[j] <- theta[j]

}

for(k in 2:N){

theta[k]~dnorm(mu.theta[k], prec.theta[k])

mu.theta[k]<- theta[k-1] + delta[k-1]

delta[k]~dnorm(delta[k-1], prec.delta[k])

prec.theta[k]<- 1/(W * sigma * sigma)

prec.delta[k]<- 1/(W * sigma * sigma)

}

theta[1]~dnorm(0,1.0E-4)

delta[1]~dnorm(0, 1.0E-4)

beta ~ dnorm(0,1.0E-4)

sigma2inv ~ dgamma(0.1, 0.1)

sigma <- 1/sqrt(sigma2inv)

W ~ dunif(0.01,100)

}

;

run;

*モデルの保存;

data _null_;

  set model;

  file "C:\WinBUGS\NDLMmodel.txt";

  put model;

run;

data _null_;

  file "C:\program files\WinBUGS14\NDLMBatch.txt";

  put // @@

  #1  "display('log')"

  #2  "check('C:/WinBUGS\NDLMmodel.txt')"

  #3  "data('C:/WinBUGS\Data.txt')"

  #4  "compile(3)"

  #5  "gen.inits()"

  #6  "update(1000)"

  #7  "thin.updater(20)"

  #8  "set(mu)"

  #9  "set(W)"

  #10 "set(delta)"

  #11 "set(sigma2inv)"

  #12 "dic.set()"

  #13 "update(1000)"

  #14 "dic.stats()"

  #15 "coda(*,'C:/WinBUGS\MCMCout')"

  #16 "save('C:/WinBUGS\log.odc')"

  #17 "quit()"

  ;

run;

*バッチファイル保存;

data _null_;

  file "C:\WinBUGS\NDLM.bat";

  put  "CD C:\program files\WinBUGS14";

  put "WinBUGS14.exe /PAR NDLMBatch.txt";

  put "exit";

run;

%dtrans(data=data, outfile=C:\WinBUGS\Data.txt, var=y);

data  _null_;

  x "C:\WinBUGS\NDLM.bat";

run; 

data log;

  infile "c:\winbugs\log.txt" truncover;

  input log $80.;

  log=translate(log," ","09"x);

run;

proc print data=log;run;

%codatrans(codapath=C:\WinBUGS, prefix=mcmcout, outcoda=coda, outstats=stat, outnums=3, outgraph=N);


Appendix D   NDLMのモンテカルロシミュレーション  
	*モデルの格納をしておく;

data model;

  input model $80.;

  cards;

model{

for(j in 1:N){

y[j] ~ dnorm(mu[j], sigma2inv)

mu[j] <- theta[j]

}

for(k in 2:N){

theta[k]~dnorm(mu.theta[k], prec.theta[k])

mu.theta[k]<- theta[k-1] + delta[k-1]

delta[k]~dnorm(delta[k-1], prec.delta[k])

prec.theta[k]<- 1/(W * sigma * sigma)

prec.delta[k]<- 1/(W * sigma * sigma)

}

theta[1]~dnorm(0,1.0E-4)

delta[1]~dnorm(0, 1.0E-4)

beta ~ dnorm(0,1.0E-4)

sigma2inv ~ dgamma(0.1, 0.1)

sigma <- 1/sqrt(sigma2inv)

W ~ dunif(0.01,100)

}

;

run;

*モデルの保存;

data _null_;

  set model;

  file "C:\WinBUGS\NDLMmodel.txt";

  put model;

run;

data _null_;

  file "C:\program files\WinBUGS14\NDLMBatch1.txt";

  put // @@

  #1  "display('log')"

  #2  "check('C:/WinBUGS\NDLMmodel.txt')"

  #3  "data('C:/WinBUGS\Data.txt')"

  #4  "compile(3)"

  #5  "gen.inits()"

  #6  "update(1000)"

  #7  "thin.updater(20)"

  #8  "set(mu)"

  #9  "set(W)"

  #10 "set(delta)"

  #11 "set(sigma2inv)"

  #12 "dic.set()"

  #13 "update(1000)"

  #14 "dic.stats()"

  #15 "coda(*,'C:/WinBUGS\MCMCout')"

  #16 "save('C:/WinBUGS\log.odc')"

  #17 "quit()"

  ;

run;

*バッチファイル保存;

data _null_;

  file "C:\WinBUGS\NDLM.bat";

  put  "CD C:\program files\WinBUGS14";

  put "WinBUGS14.exe /PAR NDLMBatch.txt";

  put "exit";

run;

%macro simulation(mc=);

  *シミュレーションデータ生成;

  %do z=1 %to &mc;

    data data(keep=y);

      seed=&z; n=8;

      do i=1 to n;

        select(i);

          when(1) y=0 + rannor(seed);

          when(2) y=1 + rannor(seed);

          when(3) y=2 + rannor(seed);

          when(4) y=3 + rannor(seed);

          when(5) y=4 + rannor(seed);

          when(6) y=3 + rannor(seed);

          when(7) y=2 + rannor(seed);

          when(8) y=1 + rannor(seed);

          otherwise;

        end;

      　output;

      end;

    run;

    %dtrans(data=data, outfile=C:\WinBUGS\Data.txt, var=y);

    data  _null_;

      x "C:\WinBUGS\NDLM.bat";

    run;

    data log;

      infile "c:\WinBUGS\log.txt" truncover;

      input log $80.;

      log=translate(log," ","09"x);

    run;

    proc print data=log;run;

    %codatrans(codapath=C:\WinBUGS, prefix=mcmcout, outcoda=coda, outstats=stat, outnums=3, outgraph=N);

    %if &z=1 %then %do;

      data allstat; delete; run;

    %end;

    data allstat;

      set allstat stat;

    run;

  %end;

  proc contents data=allstat(drop=parameter prm_grp prm_seq) out=_con noprint;

  run;

  proc sort data=_con; by varnum; run;

  proc sql noprint;

    select name into: varlist separated by ' '  from _con;

  quit;

  proc means data=allstat nway noprint;

    var &varlist;

    class parameter;

    output out=mcmean mean=&varlist; *データセットmcmeanにシミュレーション結果を出力;

  run;

%mend simulation;

*Simulation実行;

%simulation(mc=100);

_1305800467.unknown

_1307023484.unknown

_1307446708.unknown

_1308044366.unknown

_1308510735.unknown

_1308510744.unknown

_1308054458.unknown

_1308384060.unknown

_1308054165.unknown

_1307967078.unknown

_1308044354.unknown

_1307446709.unknown

_1307023699.unknown

_1307105264.unknown

_1307190525.unknown

_1307105272.unknown

_1307089559.unknown

_1307023618.unknown

_1307023641.unknown

_1307023492.unknown

_1307023067.unknown

_1307023084.unknown

_1307023380.unknown

_1307023292.unknown

_1307023075.unknown

_1305801408.unknown

_1307022646.unknown

_1305800521.unknown

_1304087933.unknown

_1305795556.unknown

_1305796936.unknown

_1305796951.unknown

_1305795575.unknown

_1304368400.unknown

_1305795528.unknown

_1304087951.unknown

_945974290.unknown

_950269894.unknown

_1304087262.unknown

_1304087715.unknown

_1303586427.unknown

_950269884.unknown

_913842765.unknown

_913842820.unknown

_913842711.unknown

